

Imagiciel de simulation d'expériences aléatoires : Schéma de Bernoulli illustré par l'expérience de la planche de Galton

Objectif :

Introduire ou illustrer la problématique du calcul des probabilités par une approche visuelle globale et appréhender "la distance" entre le modèle théorique et les observations de résultats potentiels. Faire découler les notions utiles à la construction du modèle théorique des probabilités, de réponses à des questions relatives à l'observation de résultats de l'expérience aléatoire.


Public : Classes de Premières et Terminales

Le programme de cet imagiciel propose l'observation des résultats de l'expérience suivante:

On lâche une bille qui réalise dix sauts mutuellement indépendants tels qu'à chaque saut la bille oblique vers la droite ou vers la gauche. Il y a donc onze positions envisageables à l'arrivée. La bille lancée est stockée dans un des onze tubes positionné à l'emplacement de son arrivée.


L'utilisatrice ou l'utilisateur choisit le nombre de billes à observer et fixe la valeur de la probabilité d'un saut élémentaire à droite.

Quatre modes d'observations sont alors offerts:




1. Pas à pas


Pour un petit nombre de billes, Il est utile de pouvoir s'arrêter sur chaque lancer en observant le bilan partiel. Le passage au lancer suivant s'opère avec l'appui sur [ENTER]. Le programme est suspendu avec





affichage numérique du bilan après la dernière bille.



L'appui sur [ENTER] donne alors le résultat de tous les lancers présenté sous forme de billes superposées dans les onze tubes qui recueillent les billes.




- 2.
- 3.



4. Avec trace

Pour accélérer la présentation, on automatise le passage au lancer suivant. L'arrêt entre deux lancers n'est pas possible. Les résultats partiels s'affichent au fur et à mesure. Le programme est suspendu avec affichage numérique du bilan après la dernière bille.



L'appui sur [ENTER] donne alors le résultat de tous les lancers présenté sous forme de billes superposées dans les onze tubes qui recueillent les billes.



5. Sans trace

Fonctionne comme trace, avec suppression de l'affichage numérique intermédiaire du résultat. On observe successivement les différentes trajectoires suivies par les billes. Le programme est suspendu avec affichage numérique du bilan après la dernière bille.


L'appui sur [ENTER] donne alors le résultat de tous les lancers présenté sous forme de billes superposées




dans les onze tubes qui recueillent les billes.

6. Résultat seul


Il n'y a pas d'observation des trajectoires suivies par les billes. On obtient directement le résultat de tous les lancers présenté sous forme de billes superposées dans les onze tubes qui recueillent les billes.
Exemple avec 30 lancers.



Ces observations faites, on obtient ensuite après appui sur [ENTER], une représentation graphique simultanée du résultat de la simulation et du modèle théorique (Schéma de Bernouilli calculé avec les paramètres choisis au départ).



10 lancers et $p=0,5$



30 lancers et $p=0,5$

A ce niveau du programme, l'appui sur [ENTER] ramène au début du programme avec possibilité de quitter (Mode observation sur "fin") ou de recommencer avec d'éventuels nouveaux paramètres. Un double appui sur [ENTER], relance automatiquement l'observation d'une nouvelle série de lancers avec les mêmes paramètres.

Le début du programme montre un petit texte sur la vie de Galton qui permet de patienter pendant que se font les calculs du programme.

```

F1 F2 F3 F4 F5 F6 F7
Zoom Trace ReGraph Math Draw
Un peu d'histoire ...
Francis GALTON (1822-1911) est le
cousin de Charles DARWIN. Son esprit
est ouvert à tout : Géographie, Météo-
rologie, Anthropologie. Il entre à la
Royal Society en 1853. Pionnier en sta-
tistique, il introduit le coefficient
de corrélation et même un test pour éva-
luer l'efficacité de la prière ! Il tra-
vaille avec K. PEARSON à Biometrika.
PERSO RAD AUTO FUNC
    
```

Listing du programme:

Version 89

```
Galton ()
Prgm
Local m,temp,m0
@--fenêtre de tracé ---
-10.5→xmin:10.5→xmax:1→xsc1:-1→ymin:12→ymax:1→ysc1:2→xres:setGraph("axes","off"):Zoom
Sto
PlotsOff

@--saisie des paramètres---
Dialog
Title "Planche de Galton"
Text "Nombre de jets demandés"
Request "donner un entier",nb
Request "probabilité à dte",p
DropDown "observation:",{"pas à pas(n≤10)","avec trace(n≤20)","sans
trace(n>20)","résultat (n>25)","fin"},rep
EndDlog
If rep=5
Goto fin

@--début---
expr(nb)→nb:expr(p)→p

@--points de la grille--
@[[0][11]]→m
@For j,1,10
@augment(m,{seq(i,i,-j,j,2),seq(11-j,i,-j,j,2)})→m
@EndFor
@mat▶list(m[1])→xtri
@mat▶list(m[2])→ytri
@NewPlot 1,1,xtri,ytri,,,,3

@--trajectoires-----
randMat(1,11)→res:Fill 0,res:res→m0
RandSeed rand(1000)
For j,1,nb
cumSum(seq(when(rand()<p,1,-1),i,1,10))→ab
augment({0},ab)→ab
m0→temp:1→temp[1,ab[11]/2+6]
res+temp→res
seq(11-i,i,0,10)→y1
NewPlot 2,2,ab,y1,,,,1

@--cas 1-2-3---
If rep≠4 Then
ZoomRcl:RclPic galt_im
PtText "bille n°"&string(j),-10.25,11.5
PtText "sur "&string(nb)&" jets",-10.25,10

@--résultats partiels-1-2---
If rep=1 or rep=2 Then
For i,1,11
PtText string(res[1,i]),-12.2+2*i,0
EndFor
EndIf
EndIf

@--arrêt du pas à pas-1---
If rep=1
```

```

Pause
EndFor

If rep=3 Then
For i,1,11
PtText string(res[1,i]),-12.2+2*i,0
EndFor
EndIf

@--arrêt après observation---
If rep#4
Pause

@--visualisation du résultat final dans les tubes---
{}→x3
For j,1,11
augment(x3,seq(-12+2*j,i,1,res[1,j]))→x3
EndFor
{}→h3
For j,1,11
augment(h3,seq(i,i,1,res[1,j]))→h3
EndFor
max(ymax,max(mat▶list(res)))→ymax
NewPlot 3,1,x3,h3,,1:PlotsOff 1,2

@--tubes et résultats numériques---
For i,0,9
Px1Vert 13+15*i
EndFor
Px1Horz 68
For i,1,11
PtText string(res[1,i]),-12.2+2*i,0
EndFor
Pause

@--comparaison avec théorie---
mat▶list(res)→res
seq(2*k,k,-5,5)→x4
seq(nCr(10,5+k)*p^(5+k)*(1-p)^(5-k)*nb,k,-5,5)→f4
NewPlot 1,1,x4,res,,2
NewPlot 2,4,x4,,f4,,0.5
PlotsOff 3:ZoomData

@--précisions textes sur graphique ---
If ymax<10 Then
Px1Vert 147
For i,0,floor((ymax-ymin)/ysc1)
PtText string(floor(ymax-i*ysc1)),10.7,floor(ymax-i*ysc1)+0.6*ysc1
EndFor
PtText "obsvtn (x) / mod_theo (n)",-12,-0.27*ysc1
EndIf
Pause
PlotsOff
@--retour au début---
string(nb)→nb:string(p)→p:galton()
Lbl fin
DelVar i,j,x3,h3,x4,f4,xtri,ytri,ab,y1,res,rep,nb,p
EndPrgm

```

Version 92, 92+ et V200

```
Galton ()
Prgm
Local m,temp,m0
@--fenêtre de tracé ---
-10.5→xmin:10.5→xmax:1→xcl:-1→ymin:12→ymax:1→yscl:2→xres:ZoomSto
PlotsOff

@--saisie des paramètres---
Dialog
Title "Planche de Galton"
Text "Nombre de jets demandés"
Request "donner un entier",nb
Request "probabilité (à dte)",p
DropDown "observation :",{ "pas à pas(n≤10)", "avec trace(n≤20)", "sans
trace(n>20)", "résultat seul(n>25)", "fin"},rep
EndDlog
If rep=5
Goto fin

@--début---
RclPic viegltn
expr(nb)→nb:expr(p)→p

@--points de la grille----
[[0][11]]→m
For j,1,10
augment(m,{seq(i,i,-j,j,2),seq(11-j,i,-j,j,2)})→m
EndFor
mat▶list(m[1])→xtri
mat▶list(m[2])→ytri
NewPlot 1,1,xtri,ytri,,,,3

@--trajectoires-----
randMat(1,11)→res:Fill 0,res:res→m0
RandSeed rand(1000)
For j,1,nb
cumSum(seq(when(rand()<p,1,-1),i,1,10))→ab
augment({0},ab)→ab
m0→temp:1→temp[1,ab[11]/2+6]
res+temp→res
seq(11-i,i,0,10)→yl
NewPlot 2,2,ab,yl,,,,1

@--cas 1-2-3---
If rep≠4 Then
ZoomRcl
PtText "bille n°"&string(j),-10,11
PtText "sur "&string(nb)&" jets",-10,9.5

@--résultats partiels-1-2---
If rep=1 or rep=2 Then
For i,1,11
PtText string(res[1,i]),-12.2+2*i,0
EndFor
EndIf
EndIf

@--arrêt du pas à pas-1---
If rep=1
Pause
EndFor
```

```

If rep=3 Then
For i,1,11
PtText string(res[1,i]),-12.2+2*i,0
EndFor
EndIf

©--arrêt après observation---
If rep#4
Pause

©--visualisation du résultat final dans les tubes---
{}→x3
For j,1,11
augment(x3,seq(-12+2*j,i,1,res[1,j]))→x3
EndFor
{}→h3
For j,1,11
augment(h3,seq(i,i,1,res[1,j]))→h3
EndFor
max(ymax,max(mat▶list(res)))→ymax
NewPlot 3,1,x3,h3,,1:PlotsOff 1,2

©--tubes et résultats numériques---
For i,0,9
Px1Vert 16+23*i
EndFor
Px1Horz 92
For i,1,11
PtText string(res[1,i]),-12.2+2*i,0
EndFor
Pause

©--comparaison avec théorie---
mat▶list(res)→res
seq(2*k,k,-5,5)→x4
seq(nCr(10,5+k)*p^(5+k)*(1-p)^(5-k)*nb,k,-5,5)→f4
NewPlot 1,1,x4,res,,2
NewPlot 2,4,x4,,f4,,0.5
PlotsOff 3:ZoomData

©--précisions textes sur graphique ---
If ymax<10 Then
Px1Vert 220
For i,0,floor((ymax-ymin)/ysc1)
PtText string(floor(ymax-i*ysc1)),10.7,floor(ymax-i*ysc1)+0.6*ysc1
EndFor
PtText "observation (x) / modèle proba (n)",-12,-0.3*ysc1
EndIf
Pause

©--retour au début---
string(nb)→nb:string(p)→p
galton()
Lbl fin
DelVar i,j,x3,h3,x4,f4,xtri,ytri
EndPrgm

```